Problem:
Score : 600 points

Problem Statement

There is a grid with N rows and N columns of squares. Let (i, j) be the square at the i-th row from the top and the j-th column from the left.

Each of the central (N-2) \times (N-2) squares in the grid has a black stone on it.
Each of the 2N - 1 squares on the bottom side and the right side has a white stone on it.

Q queries are given. We ask you to process them in order.
There are two kinds of queries. Their input format and description are as follows:

1 x: Place a white stone on (1, x). After that, for each black stone between (1, x) and the first white stone you hit if you go down from (1, x), replace it with a white stone.

2 x: Place a white stone on (x, 1). After that, for each black stone between (x, 1) and the first white stone you hit if you go right from (x, 1), replace it with a white stone.

How many black stones are there on the grid after processing all Q queries?

Constraints

3 \leq N \leq 2\times 10^5

0 \leq Q \leq \min(2N-4,2\times 10^5)

2 \leq x \leq N-1

Queries are pairwise distinct.

Input

Input is given from Standard Input in the following format:

N Q
Query_1
\vdots
Query_Q

Output

Print how many black stones there are on the grid after processing all Q queries.

Sample Input 1

5 5
1 3
2 3
1 4
2 2
1 2

Sample Output 1

1

After each query, the grid changes in the following way:

Sample Input 2

200000 0

Sample Output 2

39999200004

Sample Input 3

176527 15
1 81279
2 22308
2 133061
1 80744
2 44603
1 170938
2 139754
2 15220
1 172794
1 159290
2 156968
1 56426
2 77429
1 97459
2 71282

Sample Output 3

31159505795

Solution in Fortran (GNU Fortran 9.2.1) (judge verdict: Accepted)

module segment_tree_mod
    use,intrinsic :: iso_fortran_env
    implicit none

    type seg_tree
        integer(int64):: n,n2
        integer(int64),pointer:: v(:)
    contains
        procedure:: init => st_init
        procedure:: update => st_update
        procedure:: query => st_query
        procedure:: at => st_at
    end type
contains
    function f(a,b) result(ret)
        integer(int64):: a,b,ret
        ret = min(a,b)
    end function

    subroutine st_init(st, n)
        class(seg_tree):: st
        integer(int64),intent(in):: n
        integer(int64):: x
        x=1
        do while(n > x)
            x = 2*x
        end do
        st%n = x
        st%n2 = 2*x-1
        allocate(st%v(st%n2), source=10_int64**16)
    end subroutine

    subroutine st_update(st, i, x)
        class(seg_tree):: st
        integer(int64), intent(in):: i
        integer(int64):: ind
        integer(int64), intent(in):: x
        
        ind = i + st%n - 1
        st%v(ind) = x
        do while(ind > 1)
            ind = ind/2
            st%v(ind) = f(st%v(2*ind), st%v(2*ind+1))
        end do
    end subroutine

    function st_query(st, a, b) result(ret)
        class(seg_tree):: st
        integer(int64), intent(in):: a,b
        integer(int64):: ret
        ret = st_query_sub(st, a, b, 1_8, 1_8, st%n)
    end function

    function st_at(st, i) result(ret)
        class(seg_tree):: st
        integer(int64),intent(in):: i
        integer(int64):: ret
        ret = st%v(i+st%n-1)
    end function

    recursive function st_query_sub(st, a, b, k, l, r) result(ret)
        type(seg_tree),intent(in):: st
        integer(int64),intent(in):: a,b,k,l,r
        integer(int64):: ret, vl, vr

        if (r < a .or. b < l) then
            ret = 10_int64**16
        else if (a <= l .and. r <= b) then
            ret = st%v(k)
        else
            vl = st_query_sub(st,a,b,k*2, l, (l+r)/2)
            vr = st_query_sub(st,a,b,k*2+1, (l+r)/2+1, r)
            ret = f(vl,vr)
        end if
    end function
end module

program main
    use,intrinsic :: iso_fortran_env
    use segment_tree_mod
    implicit none
    integer(int64):: n,nq
    integer(int64):: i,ans,v
    integer(int64), allocatable:: q(:,:)
    type(seg_tree):: x_st, y_st

    read*, n,nq
    allocate(q(2,nq))
    do i=1,nq
        read*, q(:,i)
    end do

    ans = (n-2)*(n-2)
    call x_st%init(n)
    call y_st%init(n)
    do i=1,n-1
        call x_st%update(i,10_int64**16)
        call y_st%update(i,10_int64**16)
    end do
    call x_st%update(n,n)
    call y_st%update(n,n)

    do i=1,nq
        if (q(1,i) == 1) then
            v = x_st%query(q(2,i),n)
            ! print'("v: ",i0)', v
            ans=ans-(v-2)
            call y_st%update(v,min(y_st%at(v),q(2,i)))
        else
            v = y_st%query(q(2,i),n)
            ! print'("v: ",i0)', v
            ans=ans-(v-2)
            call x_st%update(v,min(x_st%at(v),q(2,i)))
        end if
        ! print'(i0)', ans
    end do
    print'(i0)', ans
end program main